
ABSTRACT 
Context. The problem of building a program model of a finite state machine with datapath of transitions using VHDL language 

is considered. The model synthesis process is identified with the synthesis of this type of finite state machine, since the built model 
can be used both for the analysis of the device’s behavior and for the synthesis of its logic circuit in the FPGA basis. The object of 
the research is the automated synthesis of the logic circuit of the finite state machine with the datapath of transitions, based on the 
results of which numerical characteristics of the hardware expenses for the implementation of the state machine circuit can be ob-
tained. This makes it possible to evaluate the effectiveness of using this structure of the finite state machine when implementing a 
given control algorithm. 

Objective. Development and research of a VHDL model of a finite state machine with datapath of transitions for the analysis of 
the behavior of the state machine and the quantitative assessment of hardware expenses in its logic circuit. 

Method. The research is based on the structural diagram of a finite state machine with datapath of transitions. The synthesis of 
individual blocks of the structure of the state machine is carried out according to a certain procedure by the given graph-scheme of 
the con-trol algorithm. It is proposed to present the result of the synthesis in the form of a VHDL description based on the fixed val-
ues of the states codes of the state machine. The process of synthesizing the datapath of transitions, the block of formation of codes 
of transitions operations and the block of formation of microoperations is demonstrated. VHDL description of that blocks is carried 
out in a synthesizable style, which allows synthesis of the logic circuit of the finite state machine based on FPGA with the help of 
modern CAD and obtaining numerical characteristics of the circuit, in particular, the value of hardware expenses. To analyze the 
correctness of the synthesized circuit, the process of developing the behavioral component of the VHDL model, the function of 
which is the generation of input signals of the finite state machine, is considered. The classical combination of the synthesizable and 
behavioral parts of the model allows presenting the results of the synthesis of a finite state machine with datapath of transitions as a 
separate project that can be used as a structural component of the designed digital system. 

Results. Using the example of an abstract graph-scheme of the control algorithm, a VHDL model of a finite state machine with 
datapath of transitions was developed. With the help of CAD AMD Vivado, a synthesis of the developed model was carried out and 
behavioral modeling of the operation of the finite state machine circuit was carried out. The results of the circuit synthesis made it 
possible to obtain the value of hardware expenses when implementing the circuit in the FPGA basis. According to the results of be-
havioral modeling, time diagrams were obtained, which testify to the correctness of the implementation of the functions of transitions 
and outputs of the synthesized state machine.  

Conclusions. In traditional VHDL models of finite state machines, the states do not contain specific codes and are identified us-
ing literals. This allows CAD to encode states at its own discretion. However, this approach is not suitable for describing a finite state 
machine with datapath of transitions. The transformation of states codes using a set of arithmetic and logic operations requires the 
use of fixed values of states codes, which determines the specifics of the VHDL model proposed in this paper. This and similar mod-
els can be used, in particular, in the study of the effectiveness of a finite state machine according to the criterion of hardware ex-
penses in the device circuit.  

KEYWORDS: finite state machine, datapath of transitions, VHDL model, hardware expenses, AMD Vivado CAD. 
 

ABBREVIATIONS 
CPLD is a complex programmable logic device; 
FSM is a finite state machine; 
DT is a datapath of transitions; 
GSA is a graph-scheme of algorithm; 
LUT is a look-up table; 
TO – transitions operation. 

NOMENCLATURE 
A, X, Y – sets of FSM states, logical conditions and 

microoperations accordingly; 
M, L, N – number of FSM states, logical conditions 

and microoperations accordingly; 
R – bit depth of state code; 
B – number of FSM transitions; 

O – set of transitions operations; 
I – number of transitions operations; 
RW – bit depth of code of transitions operation; 
am, K1(am), K2(am) – current state and its scalar and 

vector codes; 
as, K1(as), K2(as) – transition state and its scalar and 

vector codes; 
Xh – logical conditions that ensure the transition h; 
Yh – microoperations formed during the transition h; 
Dh – signals of code of transition state; 
Wh – signals of code of transitions operations. 
 

INTRODUCTION 
Digital systems are widely used in human activity [1]. 

One of the central units of a digital systems is a control 
unit that coordinates the functioning of all system compo-
nents [2, 3]. The control unit can be implemented in the 
form of a finite state machine (FSM), in which the control 
algorithm is implemented schematically [4, 5]. FSM can 
be implemented in the form of a Mealy FSM model or a 
Moore FSM model [2–5]. In comparison with other 
classes of control units, the FSM is characterized by 
maximum speed and maximum hardware expenses [2, 3]. 
Higher hardware expenses worsen such characteristics of 
the FSM circuit as cost, dimensions, energy consumption, 
reliability [6]. Therefore, the task of reducing hardware 
expenses in the finite state machine circuit is an important 
scientific and practical problem, forming a corresponding 
scientific direction [1–7].  

One of the FSM types is a finite state machine with 
datapath of transitions (FSM with DT). Its structure in-
cludes a special datapath that converts states codes by a 
set of operations [8]. This approach allows, under certain 
conditions, to reduce hardware expenses in comparison 
with other FSM structures. 

The design of the circuit of a digital device in the 
FPGA basis is carried out using specialized CAD based 
on the VHDL model of the device. At the moment, the 
problem of developing a VHDL model of the FSM with 
DT remains unresolved. This complicates the practical 
application of this class of finite state machines. This pa-
per proposes a solution to the problem of building a 
VHDL model of an FSM with DT given by a graph-
scheme (GSA) of control algorithm. 

The object of the study is the automated synthesis of 
the logic circuit of a finite state machine with datapath of 
transitions in CAD AMD Vivado according to a VHDL 
model that corresponds to a given GSA. 

The synthesis of a canonical finite state machine can 
be carried out in automatic mode using the XST tool built 
into CAD according to the VHDL model recommended 
by Xilinx [9]. In the case of FSM with DT, a VHDL 
model should be used, in the synthesis of which the capa-
bilities of the XST tool are not used. One of the features 
of this model is the assignment of states codes of the FSM 
in the form of binary constants. 

The subject of the study is a VHDL model of a finite 
state machine with datapath of transitions, which allows 

both the synthesis of the FSM circuit in the FPGA basis 
and the verification of the correctness of the functioning 
of the circuit by means of behavioral modeling in AMD 
Vivado CAD. 

The purpose of the work is the development and re-
search of the structure and methods of building a VHDL 
model of a finite state machine with datapath of transi-
tions with the aim of systematizing approaches to the 
automated design of this class of finite state machines in 
the FPGA basis. 

 
1 PROBLEM STATEMENT 

Let us assume that a finite state machine with datapath 
of transitions is given by the graph-scheme of the algo-
rithm G and is characterized by sets of states 
A={a1, ..., aM}, input signals X={x1, ..., xL} and microop-
erations Y={y1, ..., yN}. The design of the FSM logic cir-
cuit involves the implementation of the transition function 
T=T(X, T) and the output function Y=Y(X, T) in the FPGA 
element basis using AMD Vivado CAD (until 2023 – 
Xilinx Vivado CAD). The input data for design is the 
VHDL model of the designed device, which contains syn-
thesized and behavioral parts and allows obtaining a 
quantitative value of hardware expenses for the imple-
mentation of the circuit of the state machine in a given 
element basis. 

The work solves the problem of developing a VHDL 
model of an FSM with DT according to a given GSA and 
its investigation by means of AMD Vivado CAD. 

 
2 REVIEW OF THE LITERATURE 

In the modern theory of finite state machines, a wide 
range of methods for optimizing hardware expenses in the 
FSM circuit is known. For example, such methods are 
methods of structural decomposition [7], the essence of 
which consists in multiple transformation of logical sig-
nals, which leads to corresponding changes in the struc-
tural diagram of the FSM. 

In this article, the method of operational transforma-
tion of states codes is considered as a method of hardware 
expenses optimization [8]. According to it, the conversion 
of states codes in the system of FSM transitions is carried 
out not by means of a system of canonical Boolean equa-
tions, but by means of a set of arithmetic and logical op-
erations. Combinational circuits that implement these 
operations form the so-called datapath of transitions (DT). 
As a result, a structure of FSM with DT is formed, the 
synthesis of which is discussed in [10]. 

In paper [11], the justification of the effectiveness of 
FSM with DT in comparison with the canonical FSM 
structure according to the criterion of hardware expenses 
is presented. However, the canonical structure of FSM 
today has a rather theoretical value, while the practical 
implementation of FSM circuits is carried out with the 
help of appropriate CAD software, for example, AMD 
Vivado CAD. This is primarily due to the use of the 
FPGA element basis supported by CAD. 

Since FSM is often included in designed digital sys-
tems, support for its synthesis is implemented at the AMD 
Vivado CAD level as part of the XST tool [9]. This tool 
supports several FSM synthesis methods aimed at opti-
mizing various characteristics of the device circuit when 
implemented in the FPGA basis. Modeling the process of 
synthesizing the circuit of the state machine allows you to 
obtain the numerical values of the hardware expenses in 
the circuit of the device, expressed in the number of used 
LUT-elements. 

The XST synthesis tool, built into the AMD Vivado 
CAD, is able, under certain conditions, to find code frag-
ments in the VHDL model of the device that correspond 
to the description of the finite state machine (by state ma-
chine we mean a machine with undefined states codes). 
This process is called finite state machine extraction 
(FSM extraction). For the found state machine, the XST 
tool performs the following actions: 

– states coding according to the chosen method; 
– synthesis of the register circuit in accordance with 

the chosen method of states encoding; 
– synthesis and optimization of the circuit for transi-

tion and output functions. 
To ensure the possibility of automatic extraction of the 

state machine, in its VHDL description the following pro-
visions should be observed: 

1. The FSM states are specified in the form of a set of 
literals combined in an element of the enumerated type. 

2. The memory register must be synchronous and have 
the ability to be reset to the initial state by a Reset signal. 

3. Implementation of the transition and output func-
tion systems is realized using the case operator. 

These requirements make it possible to specify an 
FSM in the form of a VHDL model using one, two or 
three processes [9, 12–14]. Regardless of how many proc-
esses uses the state machine, the XST tool is capable of 
extracting the state machine from the VHDL code and 
coding the states according to the chosen coding method. 

The disadvantage of using the XST tool is that it is not 
possible to set specific values of states codes during the 
FSM synthesis. This makes it impossible to use optimiza-
tion methods that are based on special coding of states. 
These methods also include the method of operational 
transformation of states codes. Therefore, the XST tool 
cannot be used for the synthesis of an FSM with DT cir-
cuit. As a result, the requirements for the VHDL model of 
the FSM given in [9, 12–14] cannot be directly applied to 
the FSM with DT and need to be adjusted. 
 

3 MATERIALS AND METHODS 
The structural diagram of an FSM with DT is shown 

in Fig. 1 and contains the following blocks [10]. 

 
Figure 1 – Structural diagram of an FSM with DT 

1. Block DT realizes the following function:  
 

T = T (T, W), (1)
 

that is, converts the current state code T to a transition 
state code using a transitions operation with the W code. 

2. Block W realizes the following function: 
 

W = W (T, X), (2)
 

that is, it forms transitions operations codes that control 
the operations of the DT. 

3. Block BMO realizes the function 
 

Y = Y (X, T) (3)
 

in the case of Mealy FSM or function 
 

Y = Y (T) (4)
 

in the case of a Moore FSM, that is, it provides the im-
plementation of FSM output function. In fig. 1, the pres-
ence of a connection marked with a dashed line allows 
you to consider the structure as a Mealy FSM, the absence 
of a connection – as a Moore FSM. 

The internal structure of the DT is shown in Fig. 2 
[10]. 
 

 
Figure 2 – Internal structure of DT 

 
Blocks C1 – CI correspond to combinational circuits 

implementing a set of transitions operations (TO) 
O={O1, ..., OI}. In general, each TO can be arithmetic, 
logical or combined. When designing these blocks, if pos-
sible, each block should be optimized in order to increase 
performance and reduce hardware expenses. 

The MUX block is an R-bit multiplexer with I-
directions. Under the guidance of the W signals at the 
output of the multiplexer, the R-bit code D of the next 
FSM state is formed, which enters the input of the mem-
ory register RG. 

The RG block is an R-bit synchronous register with 
the function of resetting to the initial state by the Reset 
signal. It should be noted that in the case of FSM with 
DT, the initial state does not necessarily have a zero code. 
This register performs the function of the memory of the 
datapath of transitions and the function of the memory 
register of the finite state machine. 

Let FSM be given by GSA G (Fig. 3). This GSA is 
marked by states of Moore FSM and contains the set of 
states of the states A={a0, ..., a20} with cardinality M=21, 
the set of logical conditions X={x1, x2, x3} with cardinality 
L=3, multiple set of microoperations (output signals) 

Y={y1, ..., y7} with cardinality N=7 and B=29 FSM transi-
tions. GSA has an abstract structure and content of opera-
tor vertices and is intended to demonstrate the process of 
building a VHDL model of an FSM with DT. 

The main and most difficult stage of the synthesis of 
an FSM with DT is the so-called algebraic synthesis of 
FSM. In the process of algebraic synthesis, the following 
occurs [10]: 

1. The FSM states are matched with unique codes 
from a certain set of states codes. In the case of GSA G, 
R=5 binary digits are enough to encode M=21 states. 

2. FSM transitions correspond to certain transitions 
operations from a given set of TOs. The use of one TO for 
the implementation of several state machine transitions is 
permissible and contributes to the reduction of the total 

number of used TOs and, accordingly, to the reduction of 
hardware expenses in the FSM circuit. Those transitions 
that cannot be implemented by any of the specified TOs 
should be implemented in a canonical way using a system 
of Boolean equations. 

We will carry out an algebraic synthesis for GSA G 
under the condition that the set of transitions operations is 
formed by the following ones: O = {O1, O2, O3}: 

 

   O1: D = T + 710; (5)
   O2: D = T & 010012; (6)
   O3: D = T  000112. (7)

 
 

 
Figure 3 – Graph-scheme of algorithm G 

In these expressions, T is the code of the current FSM 
state, D is the code of the state of the transition, which is 
formed at the output of the multiplexer and enters the RG 
(Fig. 2). 

TO O1 corresponds to the operation of adding the 
decimal constant 7 to the code of the current state. The 
code of the current state T is interpreted as an unsigned 
decimal number of 5 binary digits size. The operation is 
implemented on the basis of a 5-digit binary adder, in 
which the carry from the higher bit is discarded. This is 
equivalent to the operation “(T + 7) mod 32”. For exam-
ple, (25+10) mod 32 = 3. 

TO O2 is a bitwise logical conjunction operation on 
the binary value of the current state code T and the binary 
constant 01001. 

TO O3 is a bitwise logical operation XOR on the bi-
nary value of the current state code T and the binary con-
stant 00011. 

In general, some FSM transitions can be implemented 
in a canonical way without using the specified transitions 
operations. The circuit that implements all such transi-
tions will act as a separate combinational circuit Ci as part 
of the DT (Fig. 2). In order for the multiplexer to be able 
to pass through the result of the operation of this circuit, 
we must consider it as a separate TO O4, which has its 
own code. Formally, O4 is some function  of the code of 
the current state of the automaton: 

 

O4: T =  (T). (8)
 

Successful execution of algebraic synthesis gives us a 
formal solution of the algebraic synthesis problem [10]. In 
general, there may be several formal solutions. As an ex-
ample, consider the formal solution shown in Fig. 4 (the 
method of obtaining it is not considered in this paper). 

 
Figure 4 – Formal solution of problem of algebraic synthesis of an FSM with DT for GSA G (example) 

Since the operational transformation of states codes 
does not affect the FSM output function, in Fig. 4 inside 
operational vertices, microoperations are not shown. In-
stead of them, selected decimal values of states codes and 
their binary equivalents are shown. Each FSM transition 
is marked by a transitions operation that is mapped to it. 
Operation O1 is marked with “+7”, O2 – “& 01001”, O3 – 
“ 00011” for clarity. Transitions implemented in the 
canonical way (a2a4, a5a5, a6a7) are marked with 
the symbol O4. 

Let’s explain the operational implementation of transi-
tions in Fig. 4. Transition from state a0 coded with 
K(a0)=1610=100002 to state a1 coded with 
K(a1)=1910=100112 is carried out using O3. Transition 
from state a16 coded with K(a16)=810=010002 to state a19 
coded with K(a19)=1510=011112 is carried out using O1. 
Transition from state a3 coded with K(a3)=210=000102 to 
state a5 coded with K(a5)=010=000002 is carried out using 
O2. 

Let’s encode operations of transitions O1 – O4 with 
unique binary codes of bit depth RW =  log2 4  = 2, 
formed by variables W = {w1, w2}. The result of coding is 
presented in Table 1. 
 
 
 

Table 1 – Coding of transitions operations 
Oi w1 w2 

O1 0 0 

O2 0 1 

O3 1 0 

O4 1 1 

 
Let us present the result of algebraic synthesis in the 

form of an operational table of transitions (OTT) [10], 
which for our example has the form of a Table 2. 

In the Table 4, each row corresponds to a separate 
FSM transition, the number of which is indicated in col-
umn h. The Wh column contains codes of transitions op-
erations according to the table 1. In each cell of the Wh 
column, only those variables w that are equal to 1 in the 
code of the corresponding TO, are shown. For example, 
the transition h=5 is implemented using operation O3 with 
binary code 10 (w1=1, w2=0), so in the row h=5 in the 
column Wh, only the variable w1 is indicated, which is 
equal to 1 in binary code 10. 

Let’s proceed to the construction of the VHDL model 
of the FSM with DT, the OTT of which corresponds to 
the table. 2. We will present the model in the form of syn-
thesizable and behavioral parts [5, 12, 14]. Consider the 
description of the synthesized part. 
 

Table 2 – Operational table of transitions (GSA G) 
am K1(am) K2(am) as K1(as) K2(as) Xh Wh Yh h 

a0 16 10000 a1 19 10011 1 w1 – 1 

a1 19 10011 a2 1 00001 1 w2 y1  y2 2 

a3 2 00010 x1 w1 3 a2 1 00001 

a4 24 11000 x̅1 w1 w2 

y3  y4  y5 

4 

a1 1 00001 x2 w1 5 a3 2 00010 

a5 0 00000 x̅2 w2 

y3  y6 

6 

a4 24 11000 a6 31 11111 1 – y6 7 

a8 28 11100 x3 w1 w2 8 a5 0 00000 

a12 7 00111 x̅3 – 

y1  y2 

9 

a7 21 10101 x2 w1 w2 10 a6 31 11111 

a9 6 00110 x̅2 – 

y1  y3 

11 

a7 21 10101 a8 28 11100 1 – y5 12 

a8 28 11100 a6 31 11111 1 w1 y3  y4 13 

a9 6 00110 a10 5 00101 1 w1 y4  y7 14 

a10 5 00101 a11 12 01100 1 – y2  y4  y5 15 

a16 8 01000 x3 w2 16 a11 12 01100 

a1 19 10011 x̅3 – 

y2  y5 

17 

a12 7 00111 a13 14 01110 1 – y1  y4  y6 18 

a14 13 01101 x1 w1 19 a13 14 01110 

a16 8 01000 x̅1 w2 

y7 

20 

a14 13 01101 a15 20 10100 1 – y4 21 

a5 0 00000 x2 w2 22 a15 20 10100 

a17 27 11011 x̅2 – 

y1  y3 

23 

a18 11 01011 x2 w1 24 a16 8 01000 

a19 15 01111 x̅2 – 

y3  y6 

25 

a17 27 11011 a20 9 01001 1 w2 y3  y5  y7 26 

a18 11 01011 a20 9 01001 1 w2 y1 27 

a19 15 01111 a20 9 01001 1 w2 y2  y7 28 

a20 9 01001 a0 16 10000 1 – y4  y5 29 

entity FSM is 
    generic(R: integer := 5;                 -- State code capacity 
            Rw: integer := 2;                -- W code capacity 
            L: integer := 3;                 -- Number of input signals 
            N: integer := 5);                -- Number of microoperations 
    port (X: in std_logic_vector(1 to L);    -- Input signals  
          Y: out std_logic_vector(1 to N);   -- Microoperations 
          C: in std_logic;                   -- Clock 
          Reset: in std_logic);              -- Reset 
end FSM; 

 
 

In the “generic” section, the setting constants that de-
termine the bit depth of the signal buses are defined. The 
“port” section contains the bus of input signals X, the bus 
of output microoperations Y, the synchronization signal 
Clock and the Reset signal, by which the code of the ini-
tial state of the FSM is written into the memory register. 

The architecture section contains a description of the 
internal FSM signals, as well as a description of the struc-
tural blocks in the view of processes in accordance with 
Fig. 1 and 2. The beginning of the description of the ar-
chitecture block looks like next: 

 
architecture FSM_A of FSM is 
 
signal T, D: unsigned(1 to R);        -- State code and Next state code 
signal Canonic : unsigned(1 to R);    -- Result of 'canonical' transitions 
signal nT: unsigned(1 to R);          -- Negative values of State code 
signal nX: std_logic_vector(1 to L);  -- Negative values of input signals 
signal W: std_logic_vector(1 to Rw);  -- Code of datapath operation 
 
begin 
 
nT <= not T;                           
nX <= not X; 

 
Here, the “Canonic” signal is the code of the next 

state, formed in a canonical way. Its use will be discussed 
later. 

Below a process block describing the FSM memory 
register is shown. The register switches synchronously 
with the rising edge of the Clock signal. 
 
process(C) -- Memory Register 
begin 
    if rising_edge(C) then 
        if Reset = '1' then 
            T <= "10000"; 
        else 
            T <= D; 
        end if; 
    end if; 
end process; 

 
 

The peculiarity of this description is that by a Reset 
signal equal to one, the register is transferred to the initial 
state, the code of which, according to the results of alge-
braic synthesis (Table 2), is equal to 100002. 

Let’s synthesize the block W, which forms the signals 
w1, w2 of the transitions operation code (Fig. 1). We im-
plement these signals using canonical Boolean equations 
according to the table 2 and expression (2). 

.     

;     

191817215

11331126231212

216113

982635231201

aaaxa

xaxaxaxaxaaw

xaxa

aaxaxaxaxaaw






 

 
We will use binary vectors <T1, ..., T5> to represent 

the FSM states codes. Then, according to the coding of 
the states given in the Table 2, the Boolean equations for 
signals w1, w2 take the following form: 

 

;     

     

25432115432154321

54321254321354321

254321154321543211

xTTTTTxTTTTTTTTTT

TTTTTxTTTTTxTTTTT





 

.     

254321154321543212

TTTTT

xTTTTTxTTTTTxTTTTT

xTTTTTxTTTTTTTTTTw









 

 
In general, these equations can be minimized in any 

convenient way. In this paper, we will not perform mini-
mization and will immediately present block W in the 
form of the following VHDL process: 

process (T, X, nT, nX) -- Block W 
    begin 
        W(1) <= (T(1) and nT(2) and nT(3) and nT(4) and nT(5)) or  
                (nT(1) and nT(2) and nT(3) and nT(4) and T(5)) or 
                (nT(1) and nT(2) and nT(3) and T(4) and nT(5) and X(2)) or 
                (nT(1) and nT(2) and nT(3) and nT(4) and nT(5) and X(3)) or 
                (T(1) and T(2) and T(3) and T(4) and T(5) and X(2)) or 
                (T(1) and T(2) and T(3) and nT(4) and nT(5)) or  
                (nT(1) and nT(2) and T(3) and T(4) and nT(5)) or 
                (nT(1) and T(2) and T(3) and T(4) and nT(5) and X(1)) or 
                (nT(1) and T(2) and nT(3) and nT(4) and nT(5) and X(2)); 
                      
        W(2) <= (T(1) and nT(2) and nT(3) and T(4) and T(5)) or 
                (nT(1) and nT(2) and nT(3) and nT(4) and T(5) and nX(1)) or 
                (nT(1) and nT(2) and nT(3) and T(4) and nT(5) and nX(2)) or 
                (nT(1) and nT(2) and nT(3) and nT(4) and nT(5) and X(3)) or 
                (T(1) and T(2) and T(3) and T(4) and T(5) and X(2)) or 
                (nT(1) and T(2) and T(3) and nT(4) and nT(5) and X(3)) or 
                (nT(1) and T(2) and T(3) and T(4) and nT(5) and nX(1)) or 
                (T(1) and nT(2) and T(3) and nT(4) and nT(5) and X(2)) or 
                (T(1) and T(2) and nT(3) and T(4) and T(5)) or 
                (nT(1) and T(2) and nT(3) and T(4) and T(5)) or 
                (nT(1) and T(2) and T(3) and T(4) and T(5)); 
end process; 

 
 

Please note that to ensure the correctness of the simu-
lation, the process sensitivity list contains both direct and 
inverse values of the X and T signals. 

Let’s proceed to the synthesis of the datapath of tran-
sition. Let’s clarify the structure of the DT shown in Fig. 
2, according to the results of algebraic synthesis. The 
clarification of the structure consists in the fact that it 
contains four combinational circuits C1 – C4, which corre-
spond to transitions operations O1 – O4, and the multi-
plexer is controlled by a two-bit binary code W=<w1, w2>. 

As will be shown below, combinational circuits C1 –
 C3 have a trivial implementation using operators from the 
synthesizable subset of VHDL. However, block C4 is non-
standard, as it represents a canonical implementation of a 
certain part of an FSM transitions. In the general case, the 
code of the current state T and signals of logical condi-
tions X are received at its inputs. So the refined structure 
of the OAP for GSA G is shown in Fig. 5. 

Before developing the VHDL description of the DT, 
let’s synthesize the C4 block. For this purpose, we will use 
the technique discussed in [2, 3]. 

Transitions implemented in the canonical way, in Ta-
ble 2, have numbers 4, 8 and 10. Let’s do the following. 

1. Let’s agree to use the Boolean vector <D1, ..., D5> 
to encode the transition state code. 

 
Figure 5 – Clarified structure of  DT (GSA G) 

 

2. Let’s form a separate table of transitions from these 
lines, in which instead of the column Wh there is a column 
Dh. This column indicates those components of the vector 
<D1, ..., D5> which are equal to 1 in the binary code of the 
transition state K2(as) of this row of the table. 

As a result, we will get a table of transitions imple-
mented in the canonical way (Table 3). 

Table 3 – Table of transitions implemented in the canonical way (GSA G) 
am K1(am) K2(am) as K1(as) K2(as) Xh Dh Yh h 

a2 1 00001 a4 24 11000 x̅1 D1 D2 y3  y4  y5 1 

a5 0 00000 a8 28 11100 x3 D1 D2 D3 y1  y2 2 

a6 31 11111 a7 21 10101 x2 D1 D3 D5 y1  y3 3 
 

Each of the three transitions presented in the Table 3, 
corresponds to the own term formed by the conjunction of 
signals T1, ..., T5 of the code of the current state (column 
K2(am)) and the corresponding signal of the logical condi-
tion (column Xh). Let’s form these terms: 

1543211 xTTTTTQ  ; 

3543212 xTTTTTQ  ; 

2543213 xTTTTTQ  . 

Now we can form the equation for signals D1 – D5 ac-
cording to the contents of column Dh. Since the D4 signal 
is not present in the Dh column, we will consider it always 
equal to 0. 

3211 QQQD  ; 

212 QQD  ; 

323 QQD  ; 

04 D ; 

w2 

35 QD  . 

The VHDL code for the combinational circuit C4 is 
given below. As you can see, the vector <D1, ..., D5> in 
this model corresponds to the “Canonic” signal. Although 

this signal is declared as “unsigned (1 to R) ”, each bit is 
generated separately. Also, pay attention to the presence 
in the process sensitivity list of both direct and inverse 
values of T and X signals. 

 

        process (T, X, nT, nX) -- Canonical transitions 
            variable Q1: std_logic; 
            variable Q2: std_logic; 
            variable Q3: std_logic; 
        begin 
            Q1 := nT(1) and nT(2) and nT(3) and nT(4) and T(5) and nX(1); 
            Q2 := nT(1) and nT(2) and nT(3) and nT(4) and nT(5) and X(3); 
            Q3 := T(1) and T(2) and T(3) and T(4) and T(5) and X(2); 
             
            Canonic(1) <= Q1 or Q2 or Q3; 
            Canonic(2) <= Q1 or Q2; 
            Canonic(3) <= Q2 or Q3; 
            Canonic(4) <= '0'; 
            Canonic(5) <= Q3; 
        end process; 

 
 

Although the combinational circuit C4 is described as 
a separate process, structurally it is part of the DT block 
(Fig. 5). The description of the DT block in VHDL is as 
follows: 

 

process (T, W, Canonic) -- Datapath 
begin 
    case W is 
        when "00" =>               -- O1 
            D <= T + 7; 
        when "01" =>               -- O2 
            D <= T and "01001"; 
        when "10" =>               -- O3 
            D <= T xor "00011"; 
        when "11" =>               -- O4 
            D <= Canonic; 
        when others => 
            D <= "00000"; 
    end case; 
end process; 

 

The basis of this process is the “case” operator, which 
has four branches corresponding to operation codes of 
transitions O1 – O4. Operations O1 – O4 are implemented 
with the help of “+”, “and” and “xor” operators, which 
are included in the synthesized subset of VHDL and can 
work directly with the “unsigned” data type. In the case of 
O4, to the output bus D the result obtained from the output 
of the combinational circuit C4 (input signal “Canonic”) is 
passed through. 

The following fragment of the VHDL code describes 
the BMO block that forms FSM output signals according 
to (4). The method of describing this block is not funda-
mental and can be implemented both with the help of the 
“case” operator and by setting a system of canonical Boo-
lean equations by analogy with the considered blocks W 
and C4. Also, this block can be synthesized using various 
methods of output function optimization [2–4, 7]. 
 

process (T, nT) 
begin 
    case T is 
        when "10011" => Y <= "1100000"; 
        when "00001" => Y <= "0011100"; 
        when "00010" => Y <= "0010010"; 
        when "11000" => Y <= "0000010"; 
        when "00000" => Y <= "1100000"; 
        when "11111" => Y <= "1010000"; 
        when "10101" => Y <= "0000100"; 
        when "11100" => Y <= "0011000"; 
        when "00110" => Y <= "0001001"; 
        when "00101" => Y <= "0101100"; 
        when "01100" => Y <= "0100100"; 
        when "00111" => Y <= "1001010"; 
        when "01110" => Y <= "0000001"; 
        when "01101" => Y <= "0001000"; 
        when "10100" => Y <= "1010000"; 
        when "01000" => Y <= "0010010"; 
        when "11011" => Y <= "0010101"; 
        when "01011" => Y <= "1000000"; 
        when "01111" => Y <= "0100001"; 
        when "01001" => Y <= "0001100"; 
        when others =>  Y <= "0000000";        
    end case; 
end process; 

 

The fragments of the VHDL code considered above 
form a synthesizable part of the VHDL model of the FSM 
with DT. For the correct functioning of the model, its first 
lines should be lines connecting the necessary libraries: 
 

library IEEE; 
use IEEE.STD_LOGIC_1164.ALL; 
use ieee.numeric_std.all; 

 

As the last line, you should add the architecture block 
completion statement: 
 

end FSM_A; 
 

To check the correctness of the considered VHDL 
model, it is necessary to develop its behavioral part. The 
function of the behavioral part in our case is the genera-
tion of external signals and their supply to the FSM in-
puts. Time intervals of signals generation in this case are 
of no fundamental importance, since behavioral modeling 
will take place without reference to the physical charac-
teristics of the device. 

The behavioral part can be described by the following 
VHDL code fragment: 

library IEEE; 
use IEEE.STD_LOGIC_1164.ALL; 
use ieee.numeric_std.all; 
 
entity Model is 
    generic(R: integer := 5;                        -- State code capacity 
            Rw: integer := 2;                       -- W code capacity 
            N: integer := 7;                        -- Number of microoperations 
            L: integer := 3);                       -- Number of logical conditions 
    port (Y: out std_logic_vector(1 to N));         -- Microoperations 
end Model; 
 
architecture Model_A of Model is 
signal C: std_logic;                                -- Clock     
signal Reset: std_logic;                            -- Reset 
signal X: std_logic_vector(1 to L);                 -- Logical conditions 
 
component FSM is 
    generic(R: integer := 5;                        -- State code capacity 
            Rw: integer := 2;                       -- W code capacity 
            L: integer := 3;                        -- Number of logical conditions 
            N: integer := 7);                       -- Number of microoperations 
    port (X: in std_logic_vector(1 to L);           -- Input signals  
          Y: out std_logic_vector(1 to N);          -- Microoperations 
          C: in std_logic;                          -- Clock 
          Reset: in std_logic);                     -- Reset 
end component FSM; 
 
begin 
        process                                                         -- Clock  
        begin 
           C <= '0'; wait for 80 ns; 
           C <= '1'; wait for 20 ns; 
        end process; 
 
        Reset <= '0' after 0 ns, '1' after 10 ns, '0' after 90 ns;      -- Reset 
 
        process                                                         -- X1 
        begin 
         X(1) <= '1'; wait for 17 ns; X(1) <= '0'; wait for 37 ns; 
 end process; 
      
        process                                                         -- X2 
        begin 
         X(2) <= '1'; wait for 43 ns; X(2) <= '0'; wait for 36 ns; 
 end process; 
 
        process                                                         -- X3 
        begin 
         X(3) <= '1'; wait for 38 ns; X(3) <= '0'; wait for 17 ns; 
 end process; 
 
        L1: component FSM 
            port map (X, Y, C, Reset); 
 
end Model_A; 

 
 

The behavioral part has the following features: 
1. Microoperations formed by the FSM are displayed 

on the output port Y. 
2. The Clock signal has an interval of 100 ns. The Re-

set signal is generated once at the start of the device’s 
functioning. 

3. Signals X are formed in separate processes, which 
makes them independent of each other. The intervals of 
the upper and lower levels are random and can have any 
values. 

 
 
 
 
 

4 EXPERIMENTS 
For the developed VHDL model of FSM with DT, the 

authors conducted experimental research with the help of 
CAD AMD Vivado version 2023.1 (Vivado ML Standard 
Edition, free version). The research sets two goals: 

1. Checking the correctness of the work of the synthe-
sized FSM circuit using behavioral modeling. 

2. Checking the possibility of synthesis of MPA logic 
circuit in FPGA basis. 

Achieving the first goal will allow us to consider the 
proposed approach to building a VHDL model of an FSM 
with DT correct. Achieving the second goal will confirm 
the possibility of using the developed model to evaluate 

the effectiveness of FSM with DT according to the crite-
rion of hardware expenses [11]. 

 
5 RESULTS 

Behavioral modeling of the developed model of FSM 
with DT for GSA G was performed in AMD Vivado 
CAD using standard modeling parameters. A fragment of 

the timing diagram of the state machine is shown in 
Fig. 6. Signals T and D are in unsigned decimal format, 
other signals are in binary format. Three markers are set 
on the diagram, which allow to analyze important mo-
ments of time in functioning of the FSM. Let’s consider 
them. 

 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

Figure 6 – Time diagram of functioning of FSM with DT (fragment) 
 

Until the moment t1=2650 ns, the FSM is in the state 
with code T=3110=111112 (state a6) and forms microop-
erations y1, y3. At time t1, signal x2 becomes equal to 0, 
which sets the values of signals w1 and w2 into zero val-
ues. This results to executing of operation O1: 

 

D = 3110 + 710 = (38 mod 32)10 = 610 = 001102. 
 

This value is formed on bus D and is the state code of 
the transition of the FSM in the next cycle of functioning. 

At the moment t2=2680 ns, the rising edge of the syn-
chronization signal C arrives. Following this signal, the 
value D=610=001102 is loaded into the memory register 
and appears on the bus T. Thus, the FSM correctly transi-
tioned from state a6 with code 3110 to state a9 with code 
610. Also, with a change in T, there is a change in the out-
put signals: microoperations y4, y7 are formed on the Y 
bus. This coincides with Fig. 3 and confirms the correct 
functioning of the BMO circuit. 

At the moment t3=2780 ns, the FSM after the rising 
edge of signal C passes from the state with code 
T=610=001102 (state a9) to the state with code 
D=510=001012 (state a10). The transformation of the state 
code proceeds using the operation O3: 001102  000112 = 
001012. Since this transition is unconditional, the opera-
tion code W(O3)=<10> is formed on the W bus immedi-
ately after the FSM transition to the state T=001102 (start-
ing from the moment t2) and does not change when the 
values of signals x1 – x3 change. Also, microoperations y2, 
y4, y5 are formed at time t4. This corresponds to state a10 
in Fig. 3 and to transition h=15 in table 2. 

Thus, it can be concluded that the developed VHDL 
model of FSM with DT is correct and corresponds to the 
given graph-scheme of algorithm G. 

Let’s check the possibility of synthesizing the devel-
oped VHDL model in the FPGA basis. Experiments have 
shown that stages of synthesis and implementation in the 
FPGA chip xc7a12tcpg238-1 occur without errors. As a 
result of the synthesis, the numerical values of the hard-
ware expenses for the implementation of the synthesiz-
able part of the VHDL model of the FSM with DT were 
obtained, consisting of 16 LUT elements and 5 triggers. 
Thus, the developed VHDL model can be used to evaluate 
the efficiency of the FSM circuit according to the crite-
rion of hardware expenses. 
 

6 DISCUSSION 
The method of operational transformation of states 

codes, which is the base of the structure of FSM with DT, 
provides for special coding of states of the FSM. Values 
of states codes, selected transitions operations and their 
mapping to FSM transitions form a full picture, which is 
called a formal solution to the problem of algebraic syn-
thesis of an FSM with DT. On the one hand, the special 
coding of states makes it impossible to use the finite state 
machine synthesis tools built into the XST AMD Vivado 
CAD [9]. On the other hand, knowing the specific values 
of states codes allows you to apply your own optimization 
methods aimed at optimizing hardware expenses in vari-
ous structural blocks of an FSM with DT. 

Thus, the application of structure of FSM with DT re-
quires the development of its own VHDL model, which 
differs from the models recommended by AMD Vivado 
CAD developers. The example considered in this paper 
demonstrates the following features of the approach to 
building such a model: 

1. In the FSM with DT it is allow the use of any 
arithmetic and logical operations. They can provide a dif-
ferent interpretation to the states codes of the FSM – sca-
lar (numeric) or vector (binary). Accordingly, the VHDL 
model must also support different ways of interpreting the 
states codes. In the considered example, the “unsigned” 
data type is used to represent states codes, which allows 
performing both arithmetic (scalar) and logical (vector) 
operations on states codes. If necessary, it will allow the 
use of other scalar types available in the VHDL language 
libraries. For example, to work with signed numbers, the 
“integer” data type can be used. In cases where the set of 
TOs contains only vector operations, it is sufficient to use 
only vector data type (such as “std_logic_vector”) for 
states codes. 

2. In the considered example, the combinational cir-
cuit C4 implements FSM transitions, which are imple-
mented in a canonical way according to the system of 
Boolean equations. The necessity of using such a circuit is 
due only to the results of the algebraic synthesis carried 
out by the authors for a given GSA G. In general, situa-
tions are possible when all FSM transitions are imple-
mented using a given set of TOs. In this case, there will 
be no need for a circuit like C4. 

3. The number of combinational circuit reflects the 
number of different transitions operations. In our case, it 
is a number of lines of the “case” operator, which corre-
sponds to the multiplexer in Fig. 2. In general, the smaller 
number of combinational circuit in the datapath (that is, 
the smaller number of different TOs) corresponds to de-
crease in hardware expenses in the FSM with DT circuit. 

4. Algebraic synthesis of FSM with DT demands the 
formation of set of transitions operations. Such a forma-
tion can occur both at the beginning of algebraic synthesis 
and during of its execution. When forming the set of OT, 
it is necessary to ensure that these operations can be im-
plemented with the help of operators from the synthesiz-
able subset of the VHDL language. If there is a need to 
implement a non-standard operation (scalar or vector), 
you can use an approach similar to the development of the 
C4 circuit in the above example. 

5. FSM with DT belongs to the class of FSM with 
“hardware” logic. It should be understood that, in general, 
any change in the input data (the GSA, the set of TOs, the 
bit depth of states codes, etc.) requires a complete re-
synthesis of the FSM and its VHDL model. This applies 
to the circuits of any custom digital devices. 
 

CONCLUSIONS 
The paper proposes a solution to the scientific prob-

lem of developing a VHDL model of a finite state ma-
chine with datapath of transitions. The correctness of the 
model was checked in AMD Vivado CAD. 

The scientific novelty of the work lies in the fact that 
all stages of development of a VHDL model are demon-
strated on a specific example, which allows you to under-
stand its peculiarities and differences from typical models 
of finite state machines. The main feature of the proposed 
model is that it is not focused on the use of finite state 
machine synthesis tools built into CAD and can be used 
in CADs of different FPGA manufacturers. 

The practical use of the obtained results is possible in 
the development of methods of synthesis, optimization 
and evaluation of the efficiency of a finite state machines 
with datapath of transitions, as well as other structures 
and methods aimed at optimizing the characteristics of an 
FSM circuit. 

Prospects for further research consist in solving a 
range of scientific and practical problems related to the 
development, implementation and evaluation of the effec-
tiveness of the structures and methods of synthesis of fi-
nite state machines with optimized hardware expenses. 

AНОТАЦІЯ 
Актуальність. Розглянуто задачу побудови програмної моделі мікропрограмного автомата з операційним автоматом переходів 

мовою VHDL. Процес синтезу моделі ототожнюється із синтезом даного типу автомата, оскільки побудована модель може бути 
використана як для аналізу поведінки пристрою, так і для синтезу логічної схеми в базисі FPGA. Об’єктом дослідження є автомати-
зований синтез логічної схеми мікропрограмного автомата з операційним автоматом переходів, за результатами якого можуть бути 
отримані чисельні характеристики апаратурних витрат на реалізацію схеми автомата. Це дозволяє оцінити ефективність викорис-
тання даної структури мікропрограмного автомата при реалізації заданого алгоритму керування. 

Мета. Розробка і дослідження VHDL-моделі мікропрограмного автомата з операційним автоматом переходів для аналізу пове-
дінки автомата та кількісної оцінки апаратурних витрат в його логічній схемі. 

Метод. В основу дослідження покладено структурну схему мікропрограмного автомата з операційним автоматом переходів. 
Синтез окремих блоків структури автомата здійснюється за певною процедурою відповідно до заданої граф-схеми алгоритму керу-
вання. Результат синтезу запропоновано представляти у вигляді VHDL-опису, що оснований на фіксованих значеннях кодів станів 
автомата. Продемонстрований процес синтезу операційного автомата переходів, блоку формування кодів операцій переходів та 
блоку формування мікрооперацій. VHDL-опис даних блоків здійснюється у синтезованому стилі, що дозволяє провести синтез 
логічної схеми автомата в базисі FPGA за допомогою сучасних САПР та отримати числові характеристики схеми, зокрема значення 
апаратурних витрат. Для аналізу коректності роботи синтезованої схеми розглянуто процес розробки поведінкової складової 
VHDL-моделі, функцією якої є генерація вхідних сигналів автомата. Класичне поєднання синтезованої та поведінкової частин мо-
делі дозволяє представити результати синтезу мікропрограмного автомата з операційним автоматом переходів як окремий проєкт, 
що може бути використаний в якості структурної складової проєктованої цифрової системи. 

Результати. На прикладі абстрактної граф-схеми алгоритму керування розроблено VHDL-модель мікропрограмного автомата з 
операційним автоматом переходів. За допомогою САПР AMD Vivado проведено синтез розробленої моделі та проведене поведін-
кове моделювання роботи схеми автомата. Результати синтезу схеми дозволили отримати значення апаратурних витрат при реалі-
зації схеми в базисі FPGA. За результатами поведінкового моделювання отримані диаграми часу, які свідчать про коректність реа-
лізації функцій переходів та виходів синтезованого автомата. 

Висновки. У традицйних VHDL-моделях кінцевих автоматів стани не містять конкретних кодів і ідентифікуються за допомо-
гою літералів. Це дозволяє САПР проводити кодування станів на власний розсуд. Однак такий підхід не підходить для опису мік-
ропрограмного автомата з операційним автоматом переходів. Перетворення кодів станів за допомогою множини арифметико-
логічних операцій вимагає використання фіксованих значень кодів станів, що визначає специфіку VHDL-моделі, запропонованої в 
даній роботі. Дана і подібні моделі можуть бути використані, зокрема, при дослідженні ефективності мікропрграмного автомата за 
критерієм апаратурних витрат в схемі пристрою. 

КЛЮЧОВІ СЛОВА: мікропрограмний автомат, операційний автомат переходів, VHDL-модель, апаратурні витрати, САПР 
AMD Vivado.  

